

FONCTIONS FINANCIÈRES

Les fonctions financières d'Excel permettent de calculer des mensualités, des taux d'intérêts, des durées, etc. À chaque fois, il faudra faire très attention au niveau de la durée et du taux suivant que ceux-ci sont exprimés en annuel ou en mensuel.

QUELQUES FONCTIONS FINANCIÈRES

CALCUL D'UNE ANNUALITÉ

L'annualité est le montant remboursé une seule fois par an d'après un montant emprunté sur une durée de plusieurs années et d'un taux constant :

1. Cliquer dans la cellule voulue (ici, cellule B4)
2. Onglet Formule
Financier
VPM

	A	B
1	Montant :	132 000,00 €
2	Durée en année :	1 an
3	Taux :	10,00%
4	Annualité :	

Arguments de la fonction

VPM

Taux B3 = 0,1

Npm B2 = 1

Va -B1 = -132000

Vc = nombre

Type = nombre

= 145200

Calcule le montant total de chaque remboursement périodique d'un investissement à remboursements et taux d'intérêt constants.

Va est la valeur actuelle, c'est-à-dire la valeur présente du total des remboursements futurs.

Résultat = 145 200,00 €

[Aide sur cette fonction](#) OK Annuler

- **Taux** : Cliquer dans la cellule contenant le taux (B3)
- **Npm** : cliquer dans la cellule contenant la durée (B2), c'est-à-dire le nombre de périodes.
- **Va** : Cliquer dans la cellule contenant le montant de départ en y plaçant un signe moins (-) devant pour ne pas avoir de nombres négatifs (-B1) et **OK**

Le résultat apparaît dans la cellule : il faudra rembourser 145 200,00 € par an.

PARAMÈTRES FACULTATIFS :

- **Vc** : désigne la valeur restante souhaitée (par défaut c'est 0)
 - **Type** : 1 : paiement en début de période
0 (valeur par défaut) : paiement en fin de période
- ✓ Astuces : penser à nommer les cellules contenant le taux, la durée et le montant : Cliquer dans la cellule contenant le chiffre voulu, cliquer dans la zone de nom en haut à droite, saisir un nom (exemple : Taux) et **↵**

CALCUL D'UNE MENSUALITÉ

La mensualité se calcule de la même manière que l'annuité, il faudra juste penser à diviser le taux par 12 et à multiplier la durée par 12 si ce sont des valeurs annuelles et non mensuelles.

✓ Dans l'exemple qui suit, les cellules des montants, de la durée et du taux ont été nommées.

	A	B
1	Montant :	132 000,00 €
2	Durée en année :	1 an
3	Taux :	10,00%
4	Mensualité :	

1. Cliquer dans la cellule voulue (ici, cellule B4)
2. Onglet Formule
Financier
VPM

Arguments de la fonction

VPM

Taux Taux/12 = 0,008333333

Npm Durée*12 = 12

Va -Montant = -132000

Vc = nombre

Type = nombre

= 11604,89711

Calcule le montant total de chaque remboursement périodique d'un investissement à remboursements et taux d'intérêt constants.

Va est la valeur actuelle, c'est-à-dire la valeur présente du total des remboursements futurs.

Résultat = 11 604,90 €

[Aide sur cette fonction](#) OK Annuler

- **Taux** : Cliquer dans la cellule contenant le taux (B3 qui est appelée *Taux*) et diviser par 12 pour passer d'un taux annuel à un taux mensuel (Taux/12).
- **Npm** : cliquer dans la cellule contenant la durée (B2 qui est appelée *Durée*) et multiplier par 12 pour passer d'une année à douze mois (Durée*12).
- **Va** : Cliquer dans la cellule contenant le montant de départ (B1 qui est appelé *Montant*) avec le moins devant (-Montant) et **OK**

Le résultat apparaît : il faudra rembourser 11 604,90 € par mois.

✓ S'il s'agissait d'un remboursement trimestriel, il faudrait diviser le taux par 4 et multiplier la durée par 4.

☞ Attention, la somme des douze mensualités n'est pas égale à l'annuité. Effectivement, à chaque mois, une somme est déjà remboursée donc le montant des intérêts diminue ce qui n'est pas le cas avec l'annuité, payée en une seule fois.

AUTRES FONCTIONS FINANCIÈRES

Il existe de nombreuses fonctions financières dans Excel. En voici quelques-unes pour calculer la mensualité, les intérêts, le taux, le montant de départ ou le nombre de périodes. Les fonctions qui suivent permettent, à partir de trois informations parmi les précédentes, de connaître la quatrième. Ces fonctions financières s'utilisent comme la fonction **VPM()** :

1. Cliquer dans la cellule où doit apparaître le résultat
2. Onglet Formule, Financier, Cliquer sur la fonction voulue
3. Saisir les paramètres nécessaires en passant à diviser les taux par 12 pour passer d'un taux annuel à un taux mensuel et à multiplier la durée par douze pour passer d'une année à douze mois et **OK**.

CALCUL DES INTÉRÊTS PÉRIODIQUES AVEC **INTPER()**

La fonction **Intper()** permet de connaître les intérêts versés pour chaque période de paiement en fonction du taux, de la durée et du montant emprunté :

=INTPER(Taux/12;1;Durée*12;-Montant)

- ✓ Le chiffre 1 représente la période du remboursement souhaité (1 pour le premier paiement, 2 pour le deuxième, etc.) car le montant des intérêts diminue à chaque période.

CALCUL DU TAUX AVEC **TAUX()**

La fonction **Taux()** donne le taux d'intérêt en fonction de la durée, de la mensualité et du montant emprunté (ou placé) :

=TAUX(Durée*12;Mensualité;-Montant)*12

- ✓ Éventuellement, multiplier le résultat par 12 pour avoir un taux annuel.

CALCUL DU MONTANT OBTENU AVEC **VA()**

La fonction **Va()** permet de connaître le montant emprunté en fonction du taux, de la durée et de la mensualité. Elle permet surtout de calculer le capital obtenu en plaçant une certaine somme d'argent mois par mois, an par an, etc. :

=VA(Taux/12;Durée*12;-Mensualité)

- ✓ Penser à mettre un moins (-) devant *Mensualité* pour avoir un résultat positif.

CALCUL DU NOMBRE DE PÉRIODES AVEC **NPM()**

La fonction **Npm()** affiche le nombre de périodes en fonction du taux, de la mensualité et du montant emprunté :

=NPM(Taux/12;B4;-Montant)

CALCUL DE LA PART DE REMBOURSEMENT DANS UNE MENSUALITÉ AVEC **PRINCPER()**

La fonction **Princper()** donne la part de remboursement pour chaque mensualité en fonction du taux, de la durée et du montant emprunté :

=PRINCPER(Taux/12;1;Durée*12;-Montant)

Le chiffre 1 représente la période du remboursement souhaité (1 pour le premier remboursement, 2 pour le deuxième, etc.).

EXEMPLE DE TABLEAU DE REMBOURSEMENT D'EMPRUNT

Voici un exemple de réalisation d'un tableau de remboursement d'emprunt :

PRÉPARATION

1. Réaliser le tableau des données de base (Montant, Durée en année, Taux)
2. Penser à nommer les cellules contenant les chiffres du montant emprunté, de la durée et du taux.

	A	B	C	D	E	F
1	Montant :	132 000,00 €				
2	Durée en année :	1				
3	Taux :	10,00%				
4						
5	Remboursement d'emprunts					
6						
7	Période	Montant de départ	Mensualité	Intérêts	Remboursement	Montant restant
8	1	132 000,00 €	11 604,90 €	1 100,00 €	10 504,90 €	121 495,10 €
9	=A8+1)	=F8	=VPM(Taux/12;Durée*12;-Montant)	=INTPER(Taux/12;1;Durée*12;-B9)	=C9-D9	=B9-E9
10						

RÉALISATION DU TABLEAU DE REMBOURSEMENT D'EMPRUNT

1^{ÈRE} LIGNE

- Période(A8) : saisir le chiffre 1 (1^{ère} période)
- Montant de départ (B8) : taper le signe égal et cliquer dans la cellule contenant le montant :
=Montant
- Mensualité (C8) : calculer la mensualité à partir des chiffres du tableau de base :
=VPM(Taux/12;Durée*12;-Montant)
Penser à diviser par 12 le taux et à multiplier la durée par 12 et à mettre le signe moins(-) devant le montant.
- Intérêts (D8) : calculer les intérêts à partir du tableau de base sauf pour le montant qui se calculera à partir de la colonne montant de départ (cellule B8) :
=INTPER(Taux/12;1;Durée*12;-B8)
comme la fonction utilise toujours le montant de la colonne du montant de départ (et non du tableau de base), la période est toujours égal à 1.
ou
=INTPER(Taux/12;A9;Durée*12;-Montant)
la cellule A9 désigne la période souhaitée
Le résultat de ces deux calculs *Intper()* est identique
- Remboursement (E8) : soustraire les intérêts payés de la mensualité (qui ne font pas partie des sommes remboursées) :
=C8-D8
il est également possible d'utiliser la fonction *Prncper()*
- Montant restant (F8) : soustraire du montant de la colonne du montant de départ (et non du tableau de base) le montant remboursé :
=B8-E8

2^E LIGNE

- Période (A9) : Récupérer le chiffre de la cellule précédente et ajouter 1 :
=A8+1
- Montant de départ (B8) : Récupérer le montant restant de la ligne précédente :
=F8
- Autres cellules : Pour les cellules suivantes (Mensualité, Intérêts, Remboursement et Montant restant), sélectionner les cellules de la ligne du dessus et faire une recopie vers la deuxième ligne.

AUTRES LIGNES

À présent, il est possible de faire une recopie des cellules de la deuxième ligne de calcul du tableau (ligne 9) — et non de la première — sur autant de lignes que l'on souhaite avoir de période.

Par exemple, pour un remboursement sur 10 ans, prévoir une recopie sur 120 lignes (moins la première), pour un remboursement sur 20 ans, prévoir une recopie sur 240 lignes, etc.

Toutefois, lorsque la période de remboursement est terminée, le calcul se poursuit, affichant des valeurs négatives. Pour arrêter le calcul, il faudra donc mettre une condition logique dans chacun des calculs de la deuxième ligne.

CONDITIONS LOGIQUES

1^{ère} cellule (Période)

Il faut que la période s'arrête lorsque la durée est dépassée.

Pour cela, il faut écrire la formule suivante en A9 :

=SI(A8>=Durée*12;"";A8+1)

Autres cellules

Pour les calculs des cellules suivantes de la deuxième ligne (de C9 à F9), il faut vérifier si la première cellule est vide. Si tel est le cas, il n'affiche rien, sinon, le calcul est effectué.

Il suffit donc, dans la barre des formules en haut, d'ajouter une condition logique avant le calcul déjà saisi et de fermer la parenthèse à la fin du calcul :

=Calcul devient **=si(A9="";"";Calcul)**

Exemple :

En B9 (récupération du montant précédent) :

=F8 → **=si(A9="";"";F8)**

En C9 (mensualité) :

=VPM(Taux/12;Durée*12;-Montant) → **=si(A9="";"";VPM(Taux/12;Durée*12;-Montant))**

En D9 (intérêts) :

=INTPER(Taux/12;1;Durée*12;-B8) → **=si(A9="";"";INTPER(Taux/12;1;Durée*12;-B8))**

En E9 (remboursement) :

=C8-D8 → **=si(A9="";"";C8-D8)**

En F9 (montant restant) :

=B8-E8 → **=si(A9="";"";B8-E8)**

	A	B	C	D	E	F
1	Montant :	132 000,00 €				
2	Durée en année :	1				
3	Taux :	10,00%				
4						
5	Remboursement d'emprunts					
6						
7	Période	Montant de départ	Mensualité	Intérêts	Remboursement	Montant restant
8	1	132 000,00 €	11 604,90 €	1 100,00 €	10 504,90 €	121 495,10 €
9	=SI(A8>=Durée*12;"",A8+1)	=SI(A9="";"";F8)	=SI(A9="";"";VPM(Taux/12;Durée*12;-Montant))	=SI(A9="";"";INTPER(Taux/12;1;Durée*12;-B9))	=SI(A9="";"";C9-D9)	=SI(A9="";"";B9-E9)
10						

Effectuer une recopie vers le bas sur 240 lignes (20 ans) par exemple.

- ✓ Ainsi, si le nombre d'années saisi est inférieur à 20 ans, les calculs s'arrêteront automatiquement lorsque le prêt sera remboursé.

EXERCICE TABLEAU DES INTÉRÊTS ET DE REMBOURSEMENTS

L'objectif de cet exercice est de réaliser un tableau de suivi des intérêts permettant de connaître l'intérêt payé et le remboursement à chaque période sur une durée de 10 ans.

CRÉATION DU TABLEAU

- Réaliser le tableau suivant :
- Nommer les cellules contenant le montant (Montant), la durée en année (Durée), le taux (Taux) et la mensualité (Mensualité)

	A	B	C
1	Montant :	12 000,00 €	
2	Durée en année :	6	
3	Taux :	5,00%	
4	Mensualité :		
5			
6	Calcul des intérêts et des remboursements par période		
7			
8	Période	Intérêts	Remboursements
9	1		
10			

CALCULS

MENSUALITÉ

- En cellule B4 (Mensualité), calculer la mensualité avec la fonction **Vpm()** à partir du taux, de la durée et du montant du tableau.
- Pour cette fonction et celles qui suivent, penser à transformer les chiffres annuels en chiffres mensuels.
Penser également à passer d'un montant négatif à un montant positif.

1^{ÈRE} LIGNE

- En cellule A9 (Période), saisir le chiffre 1
- En cellule B9 (Intérêts), calculer le montant des intérêts de la première période avec **Intper()**, en utilisant le taux de base, la cellule de la colonne contenant la période, la durée et le montant de base.
- En cellule C9 (Remboursement), calculer le montant des intérêts de la première période avec **Princper()**, en utilisant le taux de base, la cellule de la colonne contenant la période, la durée et le montant de base.

2^E LIGNE

- En cellule A10 (Période), mettre une condition logique, lorsque le chiffre de la période précédente est supérieur à la durée x 12, il n'affiche rien, sinon, il affiche le chiffre de la période précédente + 1
- Pour les intérêts et le remboursement, faire une recopie des cellules de la ligne du dessus
- Pour les formules financières des cellules de la deuxième ligne (B10 et C10), mettre une condition logique qui n'affiche rien si la période est vide, sinon, il effectue le calcul financier demandé.
- Effectuer la recopie de la deuxième ligne sur au moins 120 lignes pour une durée de 10 ans (10 x 12 mois).

UTILISATION


Dans le tableau d'origine, modifier la durée avec un chiffre compris entre 1 et 10 ; les calculs doivent s'arrêter à la dernière ligne de la période.

CORRECTION DU TABLEAU DES INTÉRÊTS ET DES REMBOURSEMENTS

CRÉATION DU TABLEAU

NOMS DES CELLULES

1. Cliquer dans la cellule voulue (exemple : B1)
2. Cliquer dans la zone de nom



3. Saisir le nom voulu (exemple : Montant) et

Recommencer pour les cellules de B2 de la durée en année (*Durée*), B3 du taux (*Taux*) et B4 de la mensualité (*Mensualité*)

CALCULS

1^{ÈRE} LIGNE

Mensualité

1. Cliquer dans la cellule voulue (ici, cellule B4)
2. Onglet Formule
Financier
VPM

- **Taux** : Cliquer dans la cellule contenant le taux (B3 qui est appelée *Taux*) et diviser par 12 pour passer d'un taux annuel à un taux mensuel (Taux/12).
- **Npm** : cliquer dans la cellule contenant la durée (B2 qui est appelée *Durée*) et multiplier par 12 pour passer d'une année à douze mois (Durée*12).
- **Va** : Cliquer dans la cellule contenant le montant de départ (B1 qui est appelé *Montant*) avec le moins devant (-Montant)
et

✓ =VPM(Taux/12;Durée*12;-Montant)

Intérêts

1. Cliquer dans la cellule voulue (ici, cellule A9)
2. Onglet Formule
Financier
INTPER

- **Taux** : Cliquer dans la cellule contenant le taux (B3 qui est appelée *Taux*) et diviser par 12 pour passer d'un taux annuel à un taux mensuel (Taux/12).
- **Npm** : cliquer dans la cellule contenant la durée (B2 qui est appelée *Durée*) et multiplier par 12 pour passer d'une année à douze mois (Durée*12).
- **Va** : Cliquer dans la cellule contenant le montant de départ (B1 qui est appelé *Montant*) avec le moins devant (-Montant)
et

✓ =INTPER(Taux/12;A9;Durée*12;-Montant)

Remboursements

1. Cliquer dans la cellule voulue (ici, cellule B9)

2. Onglet Formule

Financier

PRINCPER

- **Taux** : cliquer dans la cellule contenant le taux (B3 qui est appelée *Taux*) et diviser par 12 pour passer d'un taux annuel à un taux mensuel (Taux/12).
- **Pér** : cliquer dans la cellule contenant la période (ici B9)
- **Npm** : cliquer dans la cellule contenant la durée (B2 qui est appelée *Durée*) et multiplier par 12 pour passer d'une année à douze mois (Durée*12).
- **Va** : cliquer dans la cellule contenant le montant de départ (B1 qui est appelé *Montant*) avec le moins devant (-Montant) et **OK**

✓ =PRINCPER(Taux/12;A9;Durée*12;-Montant)

2^E LIGNE

Condition logique pour la période


Cliquer dans la cellule A10 et saisir la condition logique suivante :

=SI(A9>=Durée*12;"";A9+1)

Condition logique pour les calculs

1. Cliquer dans la cellule ayant le calcul (B10)

2. Dans la barre des formules, ajouter la condition logique « SI(A10="";""; » entre le signe égal et le calcul et penser à fermer une parenthèse après le calcul :



3. Recommencer pour le remboursement de la cellule C10 :

=SI(A10="";"";PRINCPER(Taux/12;A10;Durée*12;-Montant))

4. Sélectionner la deuxième ligne et faire une recopie vers le bas jusqu'environ la 250^e ligne.

UTILISATION (ET VÉRIFICATION)

Dans le tableau d'origine, dans la cellule B2, saisir une durée comprise entre 1 et 10 ans et vérifier que les calculs s'arrêtent à la dernière ligne de la période :

	A	B	C
1	Montant :	12 000,00 €	
2	Durée en année :	6 ans	
3	Taux :	5,00%	
4	Mensualité :	193,26 €	
5			
6	Calcul des intérêts et des remboursements par période		
7			
8	Période	Intérêts	Remboursements
9	1	50,00 €	143,26 €
10	2	49,40 €	143,86 €

193,26 €

✓ La somme de chaque ligne d'intérêt (50 €) et de remboursement (143,26 €) est égale à la mensualité (193,26 €). ■